



University of Pisa
MSc in Computer Engineering
Elettronics Systems

Linear Interpolator implementation in VHDL language

Leonardo Fontanelli

Academic Year 2020/2021

Contents

1	Introduction	2
1.1	Description of the assignment	2
1.2	Example of the algorithm	2
1.3	Simplifications of the algorithm	2
2	Architecture Description	4
3	VHDL implementation	5
3.1	Linear Interpolator	5
3.2	16 bits Register	7
3.3	Combinatorial Net	9
3.4	Multiplexer 4to1	12
3.5	Flip Flop D	13
3.6	Ripple Carry Adder	14
4	Testplan and Testbench	16
4.1	Python Scripts	16
4.2	Testbench	17
4.3	Simulation	18
5	Vivado Logic Synthesis and Implementation	21
5.1	Synthesis	21
5.1.1	Timing	21
5.1.2	Critical path	22
5.1.3	Utilization	23
5.1.4	Power Consumption	23
5.1.5	Maximum Operating Frequency	24
5.2	Implementation	24
5.2.1	Timing	24
5.2.2	Utilization	25
5.2.3	Power Consumption	25
5.2.4	Maximum Operating Frequency	25
6	Conclusions	26

1 Introduction

1.1 Description of the assignment

The purpose of this project is to design a digital circuit that, having as an input a sampled signal, with a sampling period equal to T , gives as an output an interpolated version of the input signal.

Defining L as the *interpolation factor*, for every sample, the system will introduce $L - 1$ points.

$$y(nT + uT) = (y_{n+1} - y_n) * u + y_n$$

$$u = \frac{k}{L}, k \in \{0, 1, \dots, L - 1\}$$

For this project, we can consider $L = 4$ and to work with 16 bit representation. Hence, with the assumptions we have made, it is possible to achieve an interpolation, giving 1 point at a time, the system will give as output 3 points that approximate the curve between the 2 points.

1.2 Example of the algorithm

As an example to clarify the algorithm, we can compute the interpolation points of $y_n = 80$ and $y_{n+1} = 247$

$$k = 0, u = \frac{0}{4} \text{ then } y = (247 - 80) * \frac{0}{4} + 80 = 80$$

$$k = 1, u = \frac{1}{4} \text{ then } y = (247 - 80) * \frac{1}{4} + 80 = 121$$

$$k = 2, u = \frac{2}{4} \text{ then } y = (247 - 80) * \frac{2}{4} + 80 = 163$$

$$k = 3, u = \frac{3}{4} \text{ then } y = (247 - 80) * \frac{3}{4} + 80 = 205$$

As we can see from this example, the results are approximated when there are rests in the calculation. So we always obtain a positive integer as a result.

With the example, we can see that starting from 2 points (80, 247), we've obtained other 4 points 80, 121, 163, 205 that represents the interpolated points to connect the input ones.

1.3 Simplifications of the algorithm

Giving a first look to the assignment and the proposed example, it is needed a simplification in order to design in a simpler way the system. The best way in order achieve this is to have only division by 2 and 4 in the formula. Hence, the algorithm has been simplified:

$$output = (y_{n+1} - y_n) * u + y_n$$

$$output = uy_{n+1} - uy_n + y_n$$

$$output = \frac{k}{L}y_{n+1} - \frac{k}{L}y_n + y_n$$

With $k = 1, 2, 3$ the equation gives the following results:

$$k = 0, output = y_n$$

$$k = 1, output = \frac{y_{n+1}}{4} - \frac{y_n}{4} + y_n$$

$$k = 2, output = \frac{y_{n+1}}{2} - \frac{y_n}{2} + y_n$$

$$k = 3, output = \frac{3}{4}y_{n+1} - \frac{3}{4}y_n + y_n = y_{n+1} - \frac{y_{n+1}}{4} + \frac{y_n}{4}$$

It is necessary to specify, that the simplifications have been made with the purpose of working with 16 bit numbers and $L = 4$.

2 Architecture Description

The architecture of the Linear Interpolator is the following:




Figure 1: Architecture of the Linear Interpolator

- *Input*: is a 16 bit entry that takes the binary representation of the number, coming from the sampled signal.
- *Output*: is a 16 bit exit of the net that gives the interpolated points for the sampled signal.
- *Clock*: represent the input clock signal.

This net has an input the sampled signal represented on 16 bits, and gives as output the interpolated signal. Hence, given two consecutive sampled points, the system should return as output three points that connects the inputs calculated with the formula saw in the previous section.

For further details of the architecture, will be given in the next section with the VHDL implementation.

3 VHDL implementation

In this section will be analyzed the VHDL implementation in details, by showing all the code for every .vhd file generated.

3.1 Linear Interpolator

The main component of the net is the Linear Interpolator, it is composed by some other internal nets.




Figure 2: Detailed Architecture of the Linear Interpolator

- *REG_16*: it's a 16 bits register, which it's useful for storing the previous input of the net and give as output the interpolated points.
- *RC*: it's the net that calculates the outputs of the linear interpolator.
- *MUX*: this is a multiplexer that select the correct output, and it's piloted by the clock.

```
1  -----
2  --  Linear Interpolator  --
3  -----
4  library IEEE;
5  use ieee.numeric_std.all;
6  use IEEE.std_logic_1164.all;
7
8  entity LI is
9      generic (n : positive := 16);
10     port(
11         input  : in std_ulogic_vector(n-1 downto 0);
12         output  : out std_ulogic_vector(n-1 downto 0);
13         clk    : in std_ulogic
14     );
15 end LI;
16
17 architecture beh of LI is
18     -----
19
```

```

20  -- Component Declaration --
21  -----
22
23  -- This register is the one that saves the previous input
24  component REG_16 is
25  port (
26      d_reg : in std_ulogic_vector(n-1 downto 0);
27      rst   : in std_ulogic;
28      clk   : in std_ulogic;
29      q_reg : out std_ulogic_vector(n-1 downto 0)
30  );
31  end component REG_16;
32
33  -- Multiplexer that select the correct output
34  component MUX is
35  port(
36      sel : in std_ulogic_vector(1 downto 0);
37      x1  : in std_ulogic_vector(n-1 downto 0);
38      x2  : in std_ulogic_vector(n-1 downto 0);
39      x3  : in std_ulogic_vector(n-1 downto 0);
40      x4  : in std_ulogic_vector(n-1 downto 0);
41      y   : out std_ulogic_vector(n-1 downto 0)
42  );
43  end component MUX;
44
45  -- Combinatorial net that calculates the outputs.
46  -- The inputs are the current point and the previous one.
47  component RC is
48  port(
49      input : in std_ulogic_vector(n-1 downto 0);
50      old_in : in std_ulogic_vector(n-1 downto 0);
51      out0   : out std_ulogic_vector(n-1 downto 0);
52      out1   : out std_ulogic_vector(n-1 downto 0);
53      out2   : out std_ulogic_vector(n-1 downto 0);
54      out3   : out std_ulogic_vector(n-1 downto 0)
55  );
56  end component RC;
57
58  -- Declaration of the array of string type
59  type array_of_string is array (0 to 3) of std_ulogic_vector(15
60      downto 0);
61
62  -----
63  -- Signals --
64  -----
65  signal old_input : std_ulogic_vector(15 downto 0);
66  -- to store the previous input value
67  signal sel_s : std_ulogic_vector(1 downto 0) := "11";
68  -- multiplexer selector
69  signal rst : std_ulogic := '0';
70  -- Output Signal - Save the 4 outputs
71  signal out_array : array_of_string;
72
73  begin
74  -- register to update old_input
75  reg: REG_16 port map(input, rst, clk, old_input);

```

```

74      -- Calculates the output
75      res: RC port map(input, old_input, out_array(0), out_array(1),
76                        out_array(2), out_array(3));
77
78      COUNTER_P : process(clk) -- process to count clocks
79      begin
80          if(rising_edge(clk)) then
81              case sel_s is
82                  when "00" => sel_s <= "01";
83                  when "01" => sel_s <= "10";
84                  when "10" => sel_s <= "11";
85                  when "11" => sel_s <= "00";
86                  when others => sel_s <= "00";
87              end case;
88          end if;
89      end process;
90
91      -- multiplexer to select correct output
92      choose: MUX port map(sel_s, out_array(0), out_array(1),
93                            out_array(2), out_array(3), output);
94  end beh;

```

3.2 16 bits Register

As already said, this register is the one that stores the input and it is composed by 16 Flip Flop D, one for every bit.




Figure 3: 16 bits register architecture

```

1  --16 bit register
2  library IEEE;
3  use IEEE.std_logic_1164.all;
4
5  -----
6  --  Entity declaration  --
7  -----
8
9  entity REG_16 is
10     generic (n : positive := 16);
11     port (
12         d_reg : in std_ulogic_vector(n-1 downto 0);
13         rst   : in std_ulogic;
14         clk   : in std_ulogic;

```

```

15     q_reg : out std_ulogic_vector(n-1 downto 0)
16 );
17 end REG_16;
18
19 -----
20 -- Architecture definition --
21 -----
22
23 architecture str of REG_16 is
24
25     component FF_D
26     port (
27         d : in std_ulogic;
28         clk : in std_ulogic;
29         rst : in std_ulogic;
30         q : out std_ulogic
31     );
32 end component;
33
34 signal q_sig : std_ulogic_vector(n-1 downto 0);
35 signal rst_sig : std_ulogic;
36 begin
37     gen_reg:
38     for i in 0 to n-1 generate
39         ffd : FF_D port map(d_reg(i), clk, rst, q_sig(i));
40     end generate gen_reg;
41     q_reg <= q_sig;
42 end str;

```

3.3 Combinatorial Net

The Combinatorial Net is the one that computes the output of the Linear Interpolator, by using the formula computed before:

$$k = 0, output = y_n$$

$$k = 1, output = \frac{y_{n+1}}{4} - \frac{y_n}{4} + y_n$$

$$k = 2, output = \frac{y_{n+1}}{2} - \frac{y_n}{2} + y_n$$

$$k = 3, output = \frac{3}{4}y_{n+1} - \frac{3}{4}y_n + y_n = y_{n+1} - \frac{y_{n+1}}{4} + \frac{y_n}{4}$$




Figure 4: Combinatorial Net architecture

The two input signals (input and old_in, the latter one coming from the 16 bits register) are shifted in order to obtain other 4 signals that are: *i1sh2*, *i1sh4*, *i2sh2*, *i2sh4* that represent the shift of the bits (2 and 4) in order to compute the division.

Then two arrays *yn1* and *yn2* are initialized with the values that must be summed. And the summed will be provided by four Ripple Carry Adders that gives the final output of the net. The code of this component is the following:

```

1  -- Intermediate combinatorial net to calculate the four output
    values
2  library IEEE;
3  use ieee.numeric_std.all;
4  use IEEE.std_logic_1164.all;
5
6  entity RC is
7      generic (n : positive := 16);
8      port (
9          input : in std_ulogic_vector(n-1 downto 0);
10         old_in : in std_ulogic_vector(n-1 downto 0);
11         out0 : out std_ulogic_vector(n-1 downto 0);
12         out1 : out std_ulogic_vector(n-1 downto 0);
13         out2 : out std_ulogic_vector(n-1 downto 0);
14         out3 : out std_ulogic_vector(n-1 downto 0)
15     );
16  end RC;
17

```

```

18 architecture rtl of RC is
19
20     -----
21     -- Component Declaration --
22     -----
23
24     component RCA is          -- net to sum signals
25     port (
26         a : in std_ulogic_vector(n-1 downto 0);
27         b : in std_ulogic_vector(n-1 downto 0);
28         cin : in std_ulogic;
29         s : out std_ulogic_vector(n-1 downto 0);
30         cout: out std_ulogic
31     );
32 end component RCA;
33
34 -- Declaration of the array of string type
35 type array_of_string is array (0 to 3) of std_ulogic_vector(15
    downto 0);
36
37 signal out_array : array_of_string; -- to save the outputs of
    RCAs
38 signal yn1 : array_of_string;
39 signal yn2 : array_of_string;
40
41 -- signals to store shifted input
42 signal i1sh2 : std_ulogic_vector(15 downto 0); -- first
    input shifted by one position (divided by two)
43 signal i2sh2 : std_ulogic_vector(15 downto 0); -- second
    input shifted by one position
44
45 signal i1sh4 : std_ulogic_vector(15 downto 0); -- first
    input shifted by two positions (divided by four)
46 signal i2sh4 : std_ulogic_vector(15 downto 0); -- second
    input shifted by two positions
47
48 -- signals for the rests
49 signal s_cout : std_ulogic_vector(3 downto 0);
50 signal s_cin, s_cout1, s_cout2 : std_ulogic := '0';
51
52 begin
53
54     main: process(input)
55     begin
56         -- in1 and in2 shifted (to divide per two)
57         i1sh2(15) <= '0'; i1sh2(14 downto 0) <= input(15 downto 1);
58         i2sh2(15) <= '0'; i2sh2(14 downto 0) <= old_in(15 downto 1);
59
60         -- in1 and in2 double shifted (to divide per four)
61
62         i1sh4(15) <= '0'; i1sh4(14) <= '0'; i1sh4(13 downto 0) <=
            input(15 downto 2);
63         i2sh4(15) <= '0'; i2sh4(14) <= '0'; i2sh4(13 downto 0) <=
            old_in(15 downto 2);
64
65         yn2(0) <= old_in; -- yn2(0) must take the 'actual' old_in
        end process;

```

```

66
67 -- u = 0
68 yn1(0) <= "0000000000000000";
69
70 -- u = 1/4
71 yn1(1) <= i1sh4;
72 SUM1 : RCA port map(i2sh4, i2sh2, s_cin, yn2(1), s_cout1);
73
74 -- u = 2/4
75 yn1(2) <= i1sh2;
76 yn2(2) <= i2sh2;
77
78 -- u = 3/4
79 yn2(3) <= i2sh4;
80 SUM2 : RCA port map(i1sh4, i1sh2, s_cin, yn1(3), s_cout2);
81
82 -- 4 Ripple Carry Adders to generate the four output signals
83 gen_RCA:
84 for i in 0 to 3 generate
85     fin_sum : RCA port map(yn1(i), yn2(i), s_cin, out_array(i),
86                             s_cout(i));
87 end generate gen_RCA;
88
89 out0 <= out_array(0);
90 out1 <= out_array(1);
91 out2 <= out_array(2);
92 out3 <= out_array(3);
93 end rtl;

```

3.4 Multiplexer 4to1

The multiplexer is used as selector for the output, it is guided by the 2 bits input *sel*, that selects the correct output.




Figure 5: Multiplexer architecture

```

1 -- Multiplexer 4to1
2 library IEEE;

```

```

3 use IEEE.STD_LOGIC_1164.ALL;
4
5 entity MUX is
6     generic (n : positive := 16);
7     port (
8         sel : in  std_ulogic_vector(1 downto 0);
9         x1  : in  std_ulogic_vector(n-1 downto 0);
10        x2  : in  std_ulogic_vector(n-1 downto 0);
11        x3  : in  std_ulogic_vector(n-1 downto 0);
12        x4  : in  std_ulogic_vector(n-1 downto 0);
13        y   : out std_ulogic_vector(n-1 downto 0)
14    );
15
16 end MUX;
17
18 architecture beh of MUX is
19 begin
20     main: process (sel) is
21     begin
22         case sel is
23             when "00" => y <= x1;
24             when "01" => y <= x2;
25             when "10" => y <= x3;
26             when "11" => y <= x4;
27             when others => y <= "0000000000000000";
28         end case;
29     end process;
30 end beh;

```

3.5 Flip Flop D




Figure 6: Flip Flop architecture

```

1 -- Flip Flop D
2 library IEEE;
3 use IEEE.std_logic_1164.all;
4
5 entity FF_D is
6     port (

```

```

7   d : in std_ulogic;
8   clk : in std_ulogic;
9   rst : in std_ulogic;
10  q : out std_ulogic
11 );
12 end FF_D;
13
14 architecture beh of FF_D is
15 begin
16   process(clk)
17   begin
18     if rising_edge(clk) then
19       if (rst = '1') then
20         q <= '0';
21       else
22         q <= d;
23       end if;
24     end if;
25   end process;
26 end beh;

```

3.6 Ripple Carry Adder

Using a bit a bit xor, the Ripple Carry Adder gives as output the summation of the inputs a and b .




Figure 7: Ripple Carry Adder architecture

```

1  -- Ripple Carry Adder: it adds a and b std_ulogic_vectors of 16
   bits giving the sum s
2  library IEEE;
3  use IEEE.std_logic_1164.all;
4
5  entity RCA is
6    generic (n : positive := 16);
7    port (
8      a : in std_ulogic_vector(n-1 downto 0);
9      b : in std_ulogic_vector(n-1 downto 0);

```

```

10     cin : in std_ulogic;
11     s : out std_ulogic_vector(n-1 downto 0);
12     cout: out std_ulogic
13 );
14 end RCA;
15
16 architecture beh of RCA is
17
18 begin
19     combinational: process(a,b,cin)
20     variable c: std_ulogic;
21     begin
22         c := cin;
23         for i in 0 to n-1 loop
24             s(i) <= a(i) xor b(i) xor c;
25             c := (a(i) and b(i)) or (a(i) and c) or (b(i) and c);
26         end loop;
27         cout <= c;
28     end process combinational;
29 end beh;

```


4 Testplan and Testbench

In order to verify the correctness of the architecture, it has been developed 2 Python scripts and a VHDL testbench in order to compare the results and verify the results.

4.1 Python Scripts

The Python script are the *"linearInterpolator.py"* and *"inputGenerator.py"*. The first one is the implementation of the Linear Interpolator in Python and computes the points of interpolation. It takes as input a file named *"input_points.txt"*, computes the interpolation points and then plots them and saves in a file named *"output.txt"* the results. In the plot we can recognize the input points with a blue point and the output (interpolated points) with a orange "x". The following is the Python code of the script:

```
1 import matplotlib.pyplot as plt
2 import numpy as np
3
4 input_points = np.loadtxt("input_points.txt", delimiter=',', dtype=
    np.int)
5 x_coords = []
6 output = []
7 y_coords = []
8
9 L = 4 # interpolation factor
10 T = 1 # period time
11 j = 0
12 i = 0
13
14 for i in range(0, len(input_points) - 1):
15     x_coords.append(T * (i - 1))
16     k = 0 # first point
17     output.append(input_points[i])
18     y_coords.append(((i - 1) * T + (k / L) * T)) # calculate the
    coordinates of the interpolated points
19     print("Element ", j, ": ", input_points[i])
20     j += 1
21     k = 1 # second point
22     element = np.math.floor(input_points[i + 1] / 4) + np.math.
    floor(input_points[i] / 2) + np.math.floor(input_points[i] / 4)
23     output.append(element)
24     y_coords.append(((i - 1) * T + (k / L) * T)) # calculate the
    coordinates of the interpolated points
25     print("Element ", j, ": ", element)
26     j += 1
27     k = 2 # third point
28     element = np.math.floor(input_points[i+1] / 2) + np.math.floor(
    input_points[i] / 2)
29     output.append(element)
30     y_coords.append(((i - 1) * T + (k / L) * T)) # calculate the
    coordinates of the interpolated points
31     print("Element ", j, ": ", element)
32     j += 1
33     k = 3 # fourth point
```

```

34     element = np.math.floor(input_points[i] / 4) + np.math.floor(
35         input_points[i+1] / 4) + np.math.floor(input_points[i+1] / 2)
36     output.append(element)
37     y_coords.append(((i - 1) * T + (k / L) * T)) # calculate the
38     # coordinates of the interpolated points
39     print("Element ", j, ": ", element)
40     j += 1
41 x_coords.append(T*i) # calculate the coordinates of the input
42 # points
43 plt.plot(x_coords, input_points, marker='o')
44 plt.plot(y_coords, output, marker='x')
45 plt.show()
46
47 outputFile = open("output.txt", 'w')
48 for i in range(0, len(output)):
49     outputFile.write("Point:" + str(i+1) + ':{>10} {:>10}'.format(
50         str(output[i]), str('{0:016b}'.format(output[i]))) + "\n")
51 outputFile.close()

```

The second Python script reads the *"input_points.txt"* and generates a file *"input.txt"* that will be copied in the Testbench code. The output of this code is the binary representation of the numbers in the *"input_points.txt"*.

```

1 import numpy as np
2
3 input_points = np.loadtxt("input_points.txt", delimiter=',', dtype=
4     np.int)
5 inputFile = open("input.txt", 'w')
6 inputFile.write("(")
7 for i in range(len(input_points)):
8     element = '"' + '{0:016b}'.format(input_points[i]) + '"'
9     if i != len(input_points) - 1:
10         element += ", "
11     inputFile.write(element)
12 inputFile.write(');')
13 inputFile.close()

```

4.2 Testbench

The Testbench code is the following, it has in the *s_input* signal the input generated by the Python script presented before. This signal is the input of the Linear Interpolator, and they are presented to the net every 4 clock cycle.

```

1 -----
2 --           Testbench           --
3 -----
4 library IEEE;
5 use IEEE.std_logic_1164.all;
6
7 entity LI_tb is
8 end LI_tb;
9
10 -----
11 -- Architecture definition --
12 -----

```

```

13
14 architecture test of LI_tb is
15
16     component LI is
17         generic(n : positive := 16);
18         port (
19             input : in std_ulogic_vector(n-1 downto 0);
20
21             output : out std_ulogic_vector(n-1 downto 0);
22             clk    : in std_ulogic
23         );
24     end component;
25
26     -----
27     --                Signals                --
28     -----
29     constant clk_p : time := 8 ns; -- clock period
30
31     -- INPUT SIGNALS
32     type array_of_input is array (0 to 15) of std_ulogic_vector(15
33         downto 0);
34     signal s_input : array_of_input := ("0000000000000111", "
35         00000000000010011", "0000100000011010", "00000000010100001", "
36         00000000010001100", "00000000000100000", "00000000000101001", "
37         0000000001010111", "0000001001101011", "0000001100100100", "
38         0000000011110001", "0000000000111111", "0000000001001010", "
39         0000000001001110", "0000010101111111", "0000010101110101");
40     signal s_in : std_ulogic_vector(15 downto 0) := "
41         0000000000000111";
42     signal s_clk : std_ulogic := '0';
43     signal i : integer := 0;
44
45     -- OUTPUT SIGNAL
46     signal s_out : std_ulogic_vector(15 downto 0);
47
48     begin
49
50         iDUT : LI port map(s_in, s_out, s_clk);
51         s_clk <= not s_clk after clk_p/2;
52         i <= i+1 after 4*clk_p; --input must change every 4
53             clock signals
54         s_in <= s_input(i);
55
56     end test;

```

In order to have a correct behaviour of the net is necessary to change the input every 4 clock periods.

4.3 Simulation

In order to simulate and compare the Python output with the VHDL output, the input number will be the same.

Hence the selected input will be:

7, 19, 2074, 161, 140, 32, 41, 87, 619, 804, 241, 63, 74, 78, 1407, 1397

inserted in the *"input_points.txt"* file.

The following is a piece of simulation, performed with the Vivado software:




Figure 8: The simulation of the net

The output is the penultimate signal in the images. Then it has been performed the execution of the Python script, the output is the following:

```

Point:1      7  000000000000111
Point:2      8  000000000001000
Point:3     12  000000000001100
Point:4     14  000000000001110
Point:5     19  000000000010011
Point:6     531 0000001000010011
Point:7    1046 0000010000010110
Point:8    1559 0000011000010111
Point:9    2074 0000100000011010
Point:10   1595 0000011000111011
Point:11   1117 0000010001011101
Point:12    638 0000001001111110
Point:13    161 0000000010100001
Point:14    155 0000000010011011
Point:15    150 0000000010010110
Point:16    145 0000000010010001
Point:17    140 0000000010001100
Point:18    113 0000000001110001
Point:19     86 0000000001010110
Point:20     59 0000000000111011
Point:21     32 0000000000100000
Point:22     34 0000000000100010
Point:23     36 0000000000100100
Point:24     38 0000000000100110
Point:25     41 0000000000101001
Point:26     51 0000000000110011
Point:27     63 0000000000111111
Point:28     74 0000000000100101
Point:29     87 0000000000101011
Point:30    218 0000000001101101
Point:31    352 00000000101100000
Point:32    484 00000000111100100
Point:33    619 0000001001101011
Point:34    664 0000001010011000
Point:35    711 0000001011000111
Point:36    757 0000001011110101
Point:37    804 0000001100100100
Point:38    663 0000001010010111
Point:39    522 0000001000001010
Point:40    381 0000000010111101
Point:41    241 00000000011110001
Point:42    195 00000000011000011
Point:43    151 00000000010010111
Point:44    106 00000000001101010
Point:45     63 0000000000111111
Point:46     64 00000000001000000
Point:47     68 00000000001000100
Point:48     70 00000000001000110
Point:49     74 00000000001001010
Point:50     74 00000000001001010
Point:51     76 00000000001001100
Point:52     76 00000000001001100
Point:53     78 00000000001001110
Point:54    409 00000000110011001
Point:55    742 00000010111100110
Point:56   1073 00000100000110001
Point:57   1407 00000101011111111
Point:58   1403 00000101011110111
Point:59   1401 00000101011110011
Point:60   1398 0000010101110110

```

Figure 9: The Python Output

By comparing the binary output of the Vivado simulation and the Python script, we can clearly see the points are the same. Hence we can conclude that the results of the net are verified.

5 Vivado Logic Synthesis and Implementation

5.1 Synthesis

At first, it has been performed the synthesis with the Vivado simulator tool. The first simulation has been performed without constraints in order to check the VHDL consistency: it concluded correctly, with some warning errors.




Figure 10: The result of the simulation

The warnings that came out from the synthesis were the following:




Figure 11: Warnings from the synthesis

Those warnings indicates that some signal were declared in a process without being inserted in the sensitivity list. Given that, in the sensitivity list must be declared anything that the process needs to know about changes of, we can discard the warning given that we are not interested in the changes of the signals, the process doesn't need to be sensible to those signals.

5.1.1 Timing

For the definitive synthesis it has been added a clock constraint of $8ns$. The results for the timing of the net are the following:

Design Timing Summary			
Setup	Hold		Pulse Width
Worst Negative Slack (WNS): 6,741 ns	Worst Hold Slack (WHS): 0,154 ns	Worst Pulse Width Slack (WPWS): 3,500 ns	
Total Negative Slack (TNS): 0,000 ns	Total Hold Slack (THS): 0,000 ns	Total Pulse Width Negative Slack (TPWS): 0,000 ns	
Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	
Total Number of Endpoints: 4	Total Number of Endpoints: 4	Total Number of Endpoints: 21	
All user specified timing constraints are met.			

Figure 12: The timing results for the synthesis

Where:

- *Worst Negative Slack - WNS*: is the critical path, corresponds to the worst slack of all the timing paths for max delay analysis.
- *Total Negative Slack - TNS*: is the sum of negative slack.
- *Worst Hold Slack - WHS*: corresponds to the worst slack of all the timing paths for min delay analysis.
- *Total Hold Slack - THS*: is the sum of negative hold slack paths.
- *Worst Pulse Width Slack - WPWS*: corresponds to the worst pulse width slack for (Min low pulse width, min high pulse width, min period, max period).
- *Total Worst Pulse Width - TWPW*: is the sum of all WPWS violations.

We can notice that all the slack are positive, hence the clock period has no need to be increased and can be decreased instead.

It resulted that the minimum clock for which the slack are all greater than 0 is 2.16ms. The resulting timing summary is the following:

Design Timing Summary			
Setup	Hold		Pulse Width
Worst Negative Slack (WNS): 0,901 ns	Worst Hold Slack (WHS): 0,154 ns	Worst Pulse Width Slack (WPWS): 0,005 ns	
Total Negative Slack (TNS): 0,000 ns	Total Hold Slack (THS): 0,000 ns	Total Pulse Width Negative Slack (TPWS): 0,000 ns	
Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	
Total Number of Endpoints: 4	Total Number of Endpoints: 4	Total Number of Endpoints: 21	
All user specified timing constraints are met.			

Figure 13: The timing results for the synthesis with the new clock

5.1.2 Critical path

The critical path resulted to be the following:




Figure 14: Critical Path

In general, the delay for the path are the following:

Intra-Clock Paths - Clock - Setup													
Name	Slack	Levels	Routes	High Fanout	From	To	Total Delay	Logic Delay	Net Delay	Requirement	Source Clock	Destination Clock	Clock Uncertainty
Path 1	0.901	0	1	16	FSM_onehot_sel_s_reg[3]/C	FSM_onehot_sel_s_reg[0]/D	0.844	0.456	0.388	2.2	Clock	Clock	0.035
Path 2	0.901	0	1	16	FSM_onehot_sel_s_reg[0]/C	FSM_onehot_sel_s_reg[1]/D	0.844	0.456	0.388	2.2	Clock	Clock	0.035
Path 3	0.901	0	1	16	FSM_onehot_sel_s_reg[2]/C	FSM_onehot_sel_s_reg[3]/D	0.844	0.456	0.388	2.2	Clock	Clock	0.035
Path 4	0.955	0	1	1	FSM_onehot_sel_s_reg[1]/C	FSM_onehot_sel_s_reg[2]/D	0.790	0.456	0.334	2.2	Clock	Clock	0.035

Figure 15: Delay for every path of the net

5.1.3 Utilization

The utilization of the net resulted to be the following:




Figure 16: Synthesis Utilization

In this scheme, we can see that 33/100 of the utilization are used for the I/O since there is 16 bit for input, 16 bits for output plus the clock bit.

5.1.4 Power Consumption

Regarding the power consumption, it has been obtained the following result:




Figure 17: Synthesis Power Consumption

We can notice that the total power consumption of the net is $0.15W$, with 39% related to Dynamic power and 61% to Static power.

5.1.5 Maximum Operating Frequency

With the clock constraint equal to $2.16ms$, it can be computed the maximum operating frequency which is equal to $f = \frac{1}{T_{clock}} = \frac{1}{2.16ms} = 462.95Hz$.

5.2 Implementation

Still with the Vivado software it has been performed the implementation step. The computation performed well and gave the same warnings explained in the previous section.

5.2.1 Timing

Also with the implementation step it has been reported the timing of the net, which is the following:

Design Timing Summary		
Setup	Hold	Pulse Width
Worst Negative Slack (WNS): 0,785 ns	Worst Hold Slack (WHS): 0,239 ns	Worst Pulse Width Slack (WPWS): 0,005 ns
Total Negative Slack (TNS): 0,000 ns	Total Hold Slack (THS): 0,000 ns	Total Pulse Width Negative Slack (TPWS): 0,000 ns
Number of Failing Endpoints: 0	Number of Failing Endpoints: 0	Number of Failing Endpoints: 0
Total Number of Endpoints: 4	Total Number of Endpoints: 4	Total Number of Endpoints: 21
All user specified timing constraints are met.		

Figure 18: The timing results for the implementation

We can notice that the slack are still positive, hence we can keep the same clock as before.

5.2.2 Utilization

The utilization resulted similar to the one of the synthesis, except for the *LUT*, which is decreased to 103, with 0.59% of utilization.




Figure 19: The utilization results for the implementation

5.2.3 Power Consumption

The power summary resulted slightly different from the synthesis:




Figure 20: The power consumption results for the implementation

In this case, the *Total On-Chip Power* resulted 0.149W with 38% for the Dynamic Power and 62% for the Static Power.

5.2.4 Maximum Operating Frequency

Since no problem has been found in the timing, we can conclude that the maximum operating frequency is the same than the synthesis and it is 462,95Hz

6 Conclusions

The designed net is able to generate interpolation points giving 2 input (one actual and one old) with the following formula: $y(nT + uT) = (y_{n+1} - y_n) * u + y_n$, $u = \frac{k}{L}$, $k \in \{0, 1, \dots, L - 1\}$. All the design assignment has been successfully satisfied. For the architecture description it has been used the VHDL language. The correctness of the circuit has been tested through simulations using testbenches and Python script. It has been done the logic synthesis in order to find the maximum clock frequency and the critical path. Also the implementation of the net has been done with the Vivado software.